
Basic Usage
    

Suppose you have a users schema that is defined as below:
defmodule Accounts.User do
  schema "users" do
    field :email, :string
    field :password_hash, :string

    field :name, :string
    field :is_active, :boolean

    field :signed_up_at, :naive_datetime
  end
end

  
    
  
  Basic filtering


Suppose you wish to enable filtering by email, name, is_active, and signed_up_at. Then you can define a QueryBuilder module as below:
defmodule Accounts.User.QueryBuilder do
  use QueryElf,
    schema: Accounts.User,
    searchable_fields: [:email, :name, :is_active, :signed_up_at]
end
The schema specifies the ecto schema for which the query builder is being created. The searchable_fields specifies fields that can be filtered on. With the above definition, you can now use the query builder's capabailities as below:
User.QueryBuilder.build_query(email: "a@b.com")
|> Repo.one

User.QueryBuilder.build_query(name__in: ["Raphael Costa", "Jose Valim"])
|> Repo.all()

User.QueryBuilder.build_query(is_active: true)
|> Repo.all()

User.QueryBuilder.build_query(signed_up_at__before: ~N[2000-01-01 23:00:07])
|> Repo.all()
You can also combine multiple conditions/filters together:
# Conditions are `anded` by default
User.QueryBuilder.build_query(name__in: ["Raphael Costa", "Jose Valim"], is_active: true)
|> Repo.all()

# But you can change to using `or`
User.QueryBuilder.build_query(_or: [name: "Mr A", email: "a@b.com"])
|> Repo.all()

# Or combine them together to create a complex query
User.QueryBuilder.build_query(
  _or: [
    name: "Mr A",
    _and: [
      email__in: ["a@b.com", "a@c.com"],
      is_active: true
    ]
  ]
)
|> Repo.all()
For more information about what filters (__in, __before, etc) are defined for each field type, check the documentation for the searchable_fields option here.

  
    
  
  Basic ordering/sorting


Assuming you have the same user schema defined above, suppose you wish to allow sorting users by name, email, and signed_up_at. QueryElf can be used to achieve this as below:
defmodule Accounts.User.QueryBuilder do
  use QueryElf,
    schema: Accounts.User,
    searchable_fields: ~w[email name is_active signed_up_at]a,
    sortable_fields: ~w[email name signed_up_at]a
end
Then sorting capabilities can be included in build_query as below:
Accounts.User.QueryBuilder.build_query([],
  order: [%{field: :name, direction: :asc}, %{field: :signed_up_at, direction: :desc}]
)
|> Repo.all()


# Shorthand expression for ordering example above
Accounts.User.QueryBuilder.build_query([],
  order: [asc: :my_int, desc: :my_bool]
)
|> Repo.all()



  

    
Advanced Usage
    

Suppose you have a users schema that is defined as below:
defmodule Accounts.User do
  schema "users" do
    field :email, :string
    field :password_hash, :string

    field :name, :string
    field :is_active, :boolean

    field :signed_up_at, :naive_datetime
    field :last_logged_in_at, :naive_datetime
  end
end
Then a basic filtering and ordering query builder can be defined as:
defmodule Accounts.User.QueryBuilder do
  use QueryElf,
    schema: Accounts.User,
    searchable_fields: ~w[email name is_active signed_up_at]a,
    sortable_fields: ~w[email name signed_up_at]a
end
Where filtering is possible on email, name, is_active, and signed_up_at, and sorting is possible on email, name, and signed_up_at. If this is confusing for you, it would be a good idea to read through the basic usage documentation.

  
    
  
  Defining custom filters


Suppose you wish to implement a custom filter to identify recently active users defined as users that logged in in the last n days. You can define and use such a filter as below:
defmodule Accounts.User.QueryBuilder do
  use QueryElf,
    schema: Accounts.User,
    searchable_fields: ~w[email name is_active signed_up_at]a,
    sortable_fields: ~w[email name signed_up_at]a

  def filter(:active_in_last_days, days, _query) do
    earliest_logged_in_at =
      NaiveDateTime.add(
        NaiveDateTime.utc_now(),
        days * 24 * 3600 * -1,
        :second
      )

    dynamic([u], u.last_logged_in_at > ^earliest_logged_in_at)
  end
end

Accounts.User.QueryBuilder.build_query(active_in_last_days: 30)
|> Repo.all()
Suppose you also have a notification preferences schema and a foriegn key from users schema to the preferences schema and you want to implement custom filters to identify recently active users that are happy to receive newsletters, then that can be achieved as below:
defmodule Accounts.User do
  schema "users" do
    field :email, :string
    field :password_hash, :string

    field :name, :string
    field :is_active, :boolean

    field :signed_up_at, :naive_datetime
    field :last_logged_in_at, :naive_datetime

    has_one :notification_preferences, NotificationPreferences
  end
end

defmodule Accounts.NotificationPreferences do
  schema "notification_preferences" do
    field :send_product_updates, :boolean
    field :send_newsletters, :boolean
  end
end

defmodule Accounts.User.QueryBuilder do
  use QueryElf,
    schema: Accounts.User,
    searchable_fields: ~w[email name is_active signed_up_at]a,
    sortable_fields: ~w[email name signed_up_at]a

  def filter(:active_in_last_days, days, _query) do
    earliest_logged_in_at =
      NaiveDateTime.add(
        NaiveDateTime.utc_now(),
        days * 24 * 3600 * -1,
        :second
      )

    dynamic([u], u.last_logged_in_at > ^earliest_logged_in_at)
  end

  def filter(:accepts_newsletters, value, query) do
    {
      # Use reusable_join instead of join as explained in the next section
      join(query, assoc(r, :notification_preferences), as: :preferences),
      dynamic([preferences: p], p.send_newsletters == ^value)
    }
  end
end

Accounts.User.QueryBuilder.build_query(active_in_last_days: 30, accepts_newsletters: true)
|> Repo.all()

  
    
  
  Using reusable joins


Given the module definition of User.QueryBuilder above, if you wrote the following query:
Accounts.User.QueryBuilder.build_query(
  _or: [
    accepts_newsletters: true,
    _and: [active_in_last_days: 5, accepts_newsletters: false]
  ]
)
Then ecto with throw an error as it would result in two named joins having the same name. This is because both accepts_newsletters: true and accepts_newsletters: false add a join each between users and notification_preferences with the name preferences and ecto doesn't allow two joins with the same name.
What you really want is that a single join is added to the query as they both are essentially the same join. This is achieved by the reusable_join macro.
import QueryElf

defmodule Accounts.User.QueryBuilder do
  use QueryElf,
    schema: Accounts.User,
    searchable_fields: ~w[email name is_active signed_up_at]a,
    sortable_fields: ~w[email name signed_up_at]a

  def filter(:accepts_newsletters, value, query) do
    {
      reusable_join(query, assoc(r, :notification_preferences), as: :preferences),
      dynamic([preferences: p], p.send_newsletters == ^value)
    }
  end
end
Internally, the reusable_join macro checks if a join with the same name has already been added to the query and if so it skips adding the join. Thus, if you have two custom filters that should be using the same join, you can ensure that happens by giving the joins in the filters the same name.
For example, suppose you want to implement custom filters to allow fetching users that are happy to receive newsletters and product updates. In this case, you would need to build two different filters so that they can be used independently and also be combined. But you want to ensure that when combined, the filters use a single join instead of multiple joins on users and notification_preferences. This is achieved by using reusable_join with the same join name in both the custom filters.
defmodule Accounts.User.QueryBuilder do
  use QueryElf,
    schema: Accounts.User,
    searchable_fields: ~w[email name is_active signed_up_at]a,
    sortable_fields: ~w[email name signed_up_at]a

  def filter(:accepts_newsletters, value, query) do
    {
      reusable_join(query, assoc(r, :notification_preferences), as: :preferences),
      dynamic([preferences: p], p.send_newsletters == ^value)
    }
  end

  def filter(:accepts_product_updates, value, query) do
    {
      reusable_join(query, assoc(r, :notification_preferences), as: :preferences),
      dynamic([preferences: p], p.send_product_updates == ^value)
    }
  end
end

Accounts.User.QueryBuilder.build_query(accepts_newsletters: true)
|> Repo.all()

Accounts.User.QueryBuilder.build_query(accepts_product_updates: true)
|> Repo.all()

Accounts.User.QueryBuilder.build_query(accepts_newsletters: true, accepts_product_updates: true)
|> Repo.all()



  

    
QueryElf behaviour
    



      
Defines an Ecto query builder.
It accepts the following options:
	:schema - the Ecto.Schema for which the queries will be built (required)
	:searchable_fields - a list of fields to build default filters for. This option is simply a
shorthand syntax for using the QueryElf.Plugins.AutomaticFilters plugin with the given list
as the fields option. You should check the plugin's documentation for more details.
(default: [])
	:sortable_fields - a list of fields to build default sorters for. This option is simply a
shorthand syntax for using the QueryElf.Plugins.AutomaticSorters plugin with the given list
as the fields option. You should check the plugin's documentation for more details.
(default: [])
	:plugins - a list of plugins that can be used to increment the query builder's
functionality. See QueryElf.Plugin for more details. (default: [])


  
    
  
  Example


defmodule MyQueryBuilder do
  use QueryElf,
    schema: MySchema,
    searchable_fields: [:id, :name],
    plugins: [
      {QueryElf.Plugins.OffsetPagination, default_per_page: 10},
      MyCustomPlugin
    ]

  def filter(:my_filter, value, _query) do
    dynamic([s], s.some_field - ^value == 0)
  end
end

MyQueryBuilder.build_query(id__in: [1, 2, 3], my_filter: 10)

  
    
  
  Sharing plugins and configuration accross query builders


Sometimes you have certain plugins that you wish to always use, while allowing some degree of
fexibility to each individual query builder definition. For those scenarios, you can use something
like the following:
defmodule MyQueryElf do
  defmacro __using__(opts) do
    # Always define filters for the `id` field
    searchable_fields = (opts[:searchable_fields] || []) ++ [:id]

    # Always define a sorter for the `id` field
    sortable_fields = (opts[:sortable_fields] || []) ++ [:id]

    # Use a default per page of `20`, but allow the user to change this value
    default_per_page = opts[:default_per_page] || 20

    # Allow the user to include extra plugins
    extra_plugins = opts[:plugins] || []

    quote do
      use QueryElf,
        schema: unquote(opts[:schema]),
        plugins: [
          {QueryElf.Plugins.AutomaticFilters, fields: unquote(searchable_fields)},
          {QueryElf.Plugins.AutomaticSorters, fields: unquote(sortable_fields)},
          {QueryElf.Plugins.OffsetPagination, default_per_page: unquote(default_per_page)},
          # put any other plugins here
        ] ++ unquote(extra_plugins)
    end
  end
end

defmodule MyQueryBuilder do
  use MyQueryElf,
    schema: MySchema,
    searchable_fields: ~w[id name age is_active roles]a
end
Using this strategy you can create a re-usable set of default plugins (and plugin configurations)
that best suits your application needs, while allowing you to use QueryElf without these
defaults if you ever need to.

    Types
  


    
      
        filter()

      


    


    
      
        metadata_type()

      


    


    
      
        options()

      


    


    
      
        sort_direction()

      


    





  
    Callbacks
  


    
      
        __query_builder__(metadata_type)

      


        Should return the query builder metadata requested.



    


    
      
        base_query()

      


        Should return the base query in this query builder. This base query will be used when defining the
empty_query/0 and build_query/1 callbacks. If not defined, defaults to the supplied schema.



    


    
      
        build_query(filter)

      


        Should receive a a keyword list or a map containing parameters and use it to build a query.



    


    
      
        build_query(filter, options)

      


        Same thing as build_query/1, but also receives some options for things like pagination and
ordering.



    


    
      
        build_query(t, filter, options)

      


        The same thing as build_query/2, but instead of building a new query it receives and extends
an existing one.



    


    
      
        empty_query()

      


        Should return a query that when applied, returns nothing.



    


    
      
        filter(atom, term, t)

      


        Receives an atom representing a filter, the parameter for the aforementioned filter and an Ecto
query. Should return an Ecto dynamic or a tuple containing an Ecto query and an Ecto dynamic.



    


    
      
        sort(atom, sort_direction, term, t)

      


        Receives an atom representing a field to order by, the order direction, an extra argument to
perform the ordering, and an Ecto query. Should return the Ecto query with the appropriate sorting
options applied.



    





  
    Functions
  


    
      
        reusable_join(query, qual, bindings, expr, opts)

      


        Similar to Ecto.Query.join/{4,5}, but can be called multiple times with the same alias.



    





      


      
        Types

        


  
    
      
      Link to this type
    
    filter()


      
       
       View Source
     


  


  

      

          @type filter() :: keyword() | map()


      



  



  
    
      
      Link to this type
    
    metadata_type()


      
       
       View Source
     


  


  

      

          @type metadata_type() :: :filters


      



  



  
    
      
      Link to this type
    
    options()


      
       
       View Source
     


  


  

      

          @type options() :: keyword()


      



  



  
    
      
      Link to this type
    
    sort_direction()


      
       
       View Source
     


  


  

      

          @type sort_direction() :: :asc | :desc


      



  


        

      

      
        Callbacks

        


  
    
      
      Link to this callback
    
    __query_builder__(metadata_type)


      
       
       View Source
     


  


  

      

          @callback __query_builder__(metadata_type()) :: term()


      


Should return the query builder metadata requested.

  



  
    
      
      Link to this callback
    
    base_query()


      
       
       View Source
     


      (optional)

  


  

      

          @callback base_query() :: Ecto.Queryable.t()


      


Should return the base query in this query builder. This base query will be used when defining the
empty_query/0 and build_query/1 callbacks. If not defined, defaults to the supplied schema.
This is useful when dealing with logical deletion or other business rules that need to be followed
every time the query builder is used. Example:
defmodule UsersQueryBuilder do
  use QueryElf,
    schema: User,
    searchable_fields: [:id, :name]

  def base_query do
    from u in User, where: is_nil(u.deleted_at)
  end
end

  



  
    
      
      Link to this callback
    
    build_query(filter)


      
       
       View Source
     


  


  

      

          @callback build_query(filter()) :: Ecto.Query.t()


      


Should receive a a keyword list or a map containing parameters and use it to build a query.

  



  
    
      
      Link to this callback
    
    build_query(filter, options)


      
       
       View Source
     


  


  

      

          @callback build_query(filter(), options()) :: Ecto.Query.t()


      


Same thing as build_query/1, but also receives some options for things like pagination and
ordering.

  



  
    
      
      Link to this callback
    
    build_query(t, filter, options)


      
       
       View Source
     


  


  

      

          @callback build_query(Ecto.Query.t(), filter(), options()) :: Ecto.Query.t()


      


The same thing as build_query/2, but instead of building a new query it receives and extends
an existing one.

  



  
    
      
      Link to this callback
    
    empty_query()


      
       
       View Source
     


  


  

      

          @callback empty_query() :: Ecto.Query.t()


      


Should return a query that when applied, returns nothing.

  



  
    
      
      Link to this callback
    
    filter(atom, term, t)


      
       
       View Source
     


      (optional)

  


  

      

          @callback filter(atom(), term(), Ecto.Query.t()) ::
  Ecto.Query.dynamic_expr() | {Ecto.Query.t(), Ecto.Query.dynamic_expr()}


      


Receives an atom representing a filter, the parameter for the aforementioned filter and an Ecto
query. Should return an Ecto dynamic or a tuple containing an Ecto query and an Ecto dynamic.
Most of the filtering should happen in the returned dynamic, and returning a query should only
be used when the query needs to be extended for the dynamic to make sense. Example:
# this is a simple comparison, so just returning a dynamic will suffice
def filter(:my_filter, value, _query) do
  dynamic([s], s.some_field == ^value)
end

# this relies on a join, so the query must be extended accordingly
def filter(:my_other_filter, value, query) do
  {
    join(query, :left, [s], assoc(s, :some_relationship), as: :related),
    dynamic([related: r], r.some_field == ^value)
  }
end

  



  
    
      
      Link to this callback
    
    sort(atom, sort_direction, term, t)


      
       
       View Source
     


      (optional)

  


  

      

          @callback sort(atom(), sort_direction(), term(), Ecto.Query.t()) :: Ecto.Query.t()


      


Receives an atom representing a field to order by, the order direction, an extra argument to
perform the ordering, and an Ecto query. Should return the Ecto query with the appropriate sorting
options applied.
Example:
  def sort(:name, direction, _arg, query) do
    sort(query, [s], [{^direction, s.name}])
  end

  


        

      

      
        Functions

        


  
    
      
      Link to this macro
    
    reusable_join(query, qual, bindings, expr, opts)


      
       
       View Source
     


      (macro)

  


  

Similar to Ecto.Query.join/{4,5}, but can be called multiple times with the same alias.
Note that only the first join operation is performed, the subsequent ones that use the same alias
are just ignored. Also note that because of this behaviour, its mandatory to specify an alias when
using this function.
This is helpful when you need to perform a join while building queries one filter at a time,
because the same filter could be used multiple times or you could have multiple filters that
require the same join.
This scenario poses a problem with how the filter/3 callback work, as you need to return a
dynamic with the filtering, which means that the join must have an alias, and by default Ecto
raises an error when you add multiple joins with the same alias.
To solve this, it is recommended to use this macro instead of the default Ecto.Query.join/{4,5}.
As an added bonus, there will be only one join in the query that can be reused by multiple
filters.

  


        

      



  

    
QueryElf.Plugin behaviour
    



      
A behaviour to be followed when creating plugins that extends the query builders functionality.

      


      
        Summary


  
    Callbacks
  


    
      
        build_query(query, builder, options)

      


        Takes the query, the builder, and the options given to the build_query function build the query
as arguments. Should return the modified query.



    


    
      
        using(options)

      


        Called when QueryElf is used with the plugin.



    





      


      
        Callbacks

        


  
    
      
      Link to this callback
    
    build_query(query, builder, options)


      
       
       View Source
     


      (optional)

  


  

      

          @callback build_query(
  query :: Ecto.Query.t(),
  builder :: module(),
  options :: QueryElf.options()
) ::
  Ecto.Query.t()


      


Takes the query, the builder, and the options given to the build_query function build the query
as arguments. Should return the modified query.

  



  
    
      
      Link to this callback
    
    using(options)


      
       
       View Source
     


      (optional)

  


  

      

          @callback using(options :: keyword()) :: Macro.t()


      


Called when QueryElf is used with the plugin.
Should be used to inject required code in the builder module.

  


        

      



  

    
QueryElf.Plugins.AutomaticFilters 
    



      
Plugin for automatically defining filters for a set of fields.
It accepts the following options:
	:fields - the list of fields for which to define filters. (required)

The defined filters will vary according to the field type in the schema:
	id or binary_id:	:$FIELD - checks if the field is equal to the given value
	:$FIELD__neq - checks if the field is different from the given value
	:$FIELD__in - checks if the field is contained in the given enumerable
	:$FIELD__not_in - checks if the field is not contained in the given enumerable


	boolean:	:$FIELD - checks if the field is equal to the given value


	integer, float or decimal:	:$FIELD - checks if the field is equal to the given value
	:$FIELD__neq - checks if the field is different from the given value
	:$FIELD__in - checks if the field is contained in the given enumerable
	:$FIELD__not_in - checks if the field is not contained in the given enumerable
	:$FIELD__gt - checks if the field is greater than the given value
	:$FIELD__lt - checks if the field is lower than the given value
	:$FIELD__gte - checks if the field is greater than or equal to the given value
	:$FIELD__lte - checks if the field is lower than or equal to the given value
	:$FIELD__is_empty - checks if the field's value is nil


	string:	:$FIELD - checks if the field is equal to the given value
	:$FIELD__neq - checks if the field is different from the given value
	:$FIELD__in - checks if the field is contained in the given enumerable
	:$FIELD__not_in - checks if the field is not contained in the given enumerable
	:$FIELD__contains - checks if the field contains the given string
	:$FIELD__starts_with - checks if the field starts with the given string
	:$FIELD__ends_with - checks if the field ends with the given string


	{:array, inner_type}:	:$FIELD - checks if the field is equal to the given value
	:$FIELD__neq - checks if the field is different from the given value
	:$FIELD__includes - checks if the field includes the given value
	:$FIELD__does_not_include - checks if the field does not include the given value
	:$FIELD__is_empty - checks if the field's value is nil


	date, time, naive_datetime, datetime, time_usec, naive_datetime_usec or datetime_usec:	:$FIELD - checks if the field is equal to the given value
	:$FIELD__neq - checks if the field is different from the given value
	:$FIELD__in - checks if the field is contained in the given enumerable
	:$FIELD__not_in - checks if the field is not contained in the given enumerable
	:$FIELD__after - checks if the field occurs after the given value
	:$FIELD__before - checks if the field occurs before the given value
	:$FIELD__is_empty - checks if the field's value is nil



Any other types are simply ignored.

  
    
  
  Example


defmodule MyQueryBuilder do
  use QueryElf,
    schema: MySchema,
    plugins: [
      {QueryElf.Plugins.AutomaticFilters, fields: ~w[id name age is_active]a}
    ]
end

MyQueryBuilder.build_query(id__in: [1, 2, 3], name__starts_with: "John")

      





  

    
QueryElf.Plugins.AutomaticSorters 
    



      
Plugin for automatically defining sorters for a set of fields.
It accepts the following options:
	:fields - the list of fields for which to define sorters. (required)


  
    
  
  Example


This definition:
defmodule MyQueryBuilder do
  use QueryElf,
    schema: MySchema,
    plugins: [
      {QueryElf.Plugins.AutomaticSorters, fields: ~w[inserted_at]a}
    ]
end
is equivalent to:
defmodule MyQueryBuilder do
  use QueryElf,
    schema: MySchema

  def sort(:inserted_at, direction, _args, query) do
    case direction do
      :asc -> order_by(query, asc: :inserted_at)
      :desc -> order_by(query, desc: :inserted_at)
    end
  end
end

      





  

    
QueryElf.Plugins.OffsetPagination 
    



      
Plugin to enable limit-offset based pagination in any query builder.
It accepts the following options:
	:default_per_page - the default value of items per page (default: 25)


  
    
  
  Example


defmodule MySchemaQueryBuilder do
  use QueryElf,
    schema: MySchema,
    plugins: [QueryElf.Plugins.OffsetPagination]
end

MySchemaQueryBuilder.build_query([], page: 2, per_page: 25)

